# $Id: PKGBUILD 90016 2013-05-06 19:36:40Z foutrelis $
# Maintainer: Sergej Pupykin <pupykin.s+arch@gmail.com>
# Maintainer: dorphell <dorphell@archlinux.org>

pkgname=dillo
pkgver=3.0.3
pkgrel=2
pkgdesc="A small, fast graphical web browser built on FLTK"
arch=(i686 x86_64)
url="http://www.dillo.org"
license=('GPL')
depends=('fltk' 'libjpeg' 'perl' 'openssl' 'libpng' 'gcc-libs' 'libxcursor'
         'libxi' 'libxinerama')
backup=(etc/dillo/{dillorc,dpidrc})
source=(http://www.dillo.org/download/$pkgname-$pkgver.tar.bz2)
md5sums=('726cd0b7a18c5e25f4d80ebeffe7607e')

build() {
  cd "$srcdir/$pkgname-$pkgver"
  ./configure --prefix=/usr --sysconfdir=/etc --enable-cookies --enable-dlgui \
              --enable-ssl
  make
}

package() {
  cd "$srcdir/$pkgname-$pkgver"
  make DESTDIR="$pkgdir" install
}
